
PROJECT:

REAL CODED GENETIC ALGORITHM 

FOR BRAGG GRATING PARAMETER SYNTHESIS

The objective of this project is to study an article dealing with the synthesis of 3 parameters of a Bragg grating using a real valued genetic algorithm and to reproduce the obtained results. Compared to the work that has been already done during the course and the computer session, the direct problem resolution is different as well as the stochastic processes used in the genetic algorithm.

Main reference:  'Real-coded genetic algorithm for Bragg grating parameter synthesis' by G. Cormier and R. Boudreau, J. Opt. Soc. Am. B, Vol. 18, No 12, 2001.

Available at http://www.ann.jussieu.fr/~dumas/FBG-AG.pdf

A detailed presentation of the article is given below. The paragraph numbers are issued from the original article:

1. Introduction:
Some generalities are recalled about Fiber Bragg Gratings (FBG) and the various methods to solve the corresponding inverse problem, that is the synthesis of a FBG with a particular reflectivity spectrum. Among the works that are mentioned is the article of Skaar and Risvik (ref. 9) that has been extensively studied during the course and the computer sessions.

2. Genetic algorithm:

The principles of the real valued genetic algorithm that is used in the article are given in details in the following sub paragraphs:

A Initial population: 

The search domain consists of three real parameters, L: length of the FBG, : period of the FBG and n: apodization of the FBG

B. Selection:
The error type cost function is first given. Then a stochastic remainder selection is used. It consists in 3 steps:

(i) computation of the probability of reproduction PRj for each element j

(ii) generation of int(PRj) elements in the mating pool (int: integer part)

(iii) roulette wheel selection of the remainder of the mating pool by using the fractional parts of each element, that is PRj-int(PRj), sorted by rank 

Note that part (iii) has already been implemented in the genetic algorithms code that has been written in the computer session and that is recalled in Annex A.

C. Reproduction:
The Wright technique is well explained in the article. Note that three offsprings are generated from two parents but only the best two are kept.

D. Mutation:
The non uniform mutation that is used is rather similar to the one written in the computer sessions.

E. Elitism:
The 1-elitism that is used is rather similar to the one written in the computer sessions.

3. F-Matrix formalism

The F-Matrix method to solve the direct problem (find the reflectivity spectrum of a specific FBG) is explained. The FBG is first supposed to have a uniform apodization (n(z)=constant). In this case, formula (16) gives an expression of the reflectivity for a specific valued of  (or  equivalently). It is computed from the coefficients of a so-called transmission matrix given in (13). A coupling coefficient  is in particular given in formula (8) which can be simplified by assuming that n22+n12( n12 and n22-n12( 2n1n. In the case of a non uniform grating (n(z) no more constant), the grating is divided in m smaller gratings where the apodization is supposed to be constant. The reflectivity is thus obtained from a product of m matrices of type (13) for each small grating. 

In the limiting case of m going to infinity, such method is assumed to be equivalent to the one which has been written in the computer session based on the resolution of a Riccati type equation and which is recalled in Annex B..

4. Implementation

The details of the numerical parameters used in the genetic algorithm are given in Table 1. In all the simulations, the unknowns are L, the length of the FBG, , its period and n, its apodization.

5. Results

Various computations are done in order to validate the approach. Except for the last simulation, all the problems consist in synthesizing a particular FBG which has been first simulated. In some sense, it can be called a false inverse problem because the solution is already known. In the first computation, the apodization is supposed to be constant and the FBG has no chirp (another parameter which can be added but which will not be considered here). The results, given in Figure 2, show a very good agreement between the original and the fitted spectra. The other computation that will be studied is the fourth one (apodised and non chirped grating with 100 sections). In this case, the grating is divided in 100 small gratings where the apodisation is supposed to be constant and given by formula (18). The results, also very satisfactory, are given in Table 4, Figure 6 and Figure 7. In these computations, the exact extent of the search domain, which is not given, consists in an hypercube of R3centered in the three original values for L,  and n, actually, 10mm, 534nm and 0.00011 respectively.

Work to do: the work consists in writing a Scilab program accompanied with a small presentation (2 or 3 pages in word or Latex) reproducing the main results obtained in the article, more precisely the first and fourth computation presented in paragraph 5.

The documents must be sent at dumas@ann.jussieu.fr. You can also use this email in case of technical problems or misunderstandings during the project writing.

Deadline: April, 24th

ANNEX 1 : example of a real valued genetic algorithm for Scilab
//

// selection: proportionate with ranking method 

// crossover: blend 

// mutation: non uniform

// 1-elitism

//

function y=rastrigin(x,n)  // the function to optimize

y=n+sum(x.^2-cos(2*%pi*x));

endfunction

// 

function evalpop=evaluation(X);  // evaluation of the whole population

[Npop,n]=size(X);n=n-1;

evalpop=zeros(Npop,1);

for i=1:Npop;

  evalpop(i)=rastrigin(X(i,1:n),n);

end

endfunction

//

function [Xpar,bestpar]=evalplus1elitism(Xpar,Xoldpar,Npop,gen) 

val=evaluation(Xpar);    

Xpar(:,n+1)=val;

[minval,index]=min(Xoldpar(:,n+1));  

bestpar=Xoldpar(index(1),:);

[minval,index]=min(val);

bestnewpop=Xpar(index(1),:);

if (bestnewpop($)>bestpar($))&(gen>1) then

    nrand=int(Npop*rand())+1;

    Xpar(nrand,:)=bestpar;

else 

    bestpar=bestnewpop;

end

endfunction

//

function Xsel=rankselect(X)    // ranking selection

Xsel=X;

[Npop,n]=size(X);n=n-1;

[Fsort,ind]=sort(X(:,n+1));

X=X(ind,:);

p=1:Npop;roulette=cumsum(p)/sum(p);

for i=1:Npop;

  u=rand();

  index=find(roulette<u);Nselect=max(index)+1;

  Xsel(i,:)=X(Nselect,:);

end

endfunction

//

function xoff=blcross(xpar)    // blend crossover

[np,n]=size(xpar);n=n-1;

xoff=xpar;

for i=1:n

   u=rand();

   xoff(1,:)=u*xpar(1,:)+(1-u)*xpar(2,:);

   xoff(2,:)=u*xpar(2,:)+(1-u)*xpar(1,:);

end

endfunction

//

function xoff=mutate(xpar,ngen,Ngen,xmin,xmax) // non uniform mutation

[np,n]=size(xpar);n=n-1;b=5;xoff=xpar;

for i=1:n

  u1=rand();u2=rand();

  if (u1<1/2) then

    xoff(1,i)=xpar(1,i)+(xmax(i)-xpar(i))*u2*(1-(ngen-1)/Ngen)^b;

   else

    xoff(1,i)=xpar(1,i)-(xpar(i)-xmin(i))*u2*(1-(ngen-1)/Ngen)^b;

   end;

end;

endfunction

//

//---------main program ---------------------

//

xmin=-5.12;xmax=5.12;N=300;

x=xmin:((xmax-xmin)/(N-1)):xmax;

n=evstr(x_dialog('parameter number of the rastrigin function','2')); 

Npop=evstr(x_dialog('population number','30'));     

Ngen=evstr(x_dialog('generation number','50')); 

pc=evstr(x_dialog('crossover probability','0.9')); 

pm=evstr(x_dialog('mutation probability','0.6')); 

//

xmin=-5.12*ones(1,n);

xmax=5.12*ones(1,n);

Xmin=ones(Npop,1)*xmin;

Xmax=ones(Npop,1)*xmax;

u=rand(Npop,n);

pop=Xmin+(Xmax-Xmin).*u;  // random initialisation of the population

Xpar=[pop,zeros(Npop,1)];  

Xoff=Xpar;Xoldpar=Xpar;fmin=[];mineval=0;newval=zeros(Npop,1);Traj=[];

//

for gen=1:Ngen;

  Xpar=Xoff;

  [Xpar,bestpar]=evalplus1elitism(Xpar,Xoldpar,Npop,gen);

  Xoldpar=Xpar;

  fmin=[fmin,bestpar($)];

  //

  Xpar=rankselect(Xpar); // rank selection

  for i=1:2:(Npop-1)

    u1=int(Npop*rand())+1;u2=int(Npop*rand())+1;u3=rand();

    xpar=[Xpar(u1,:);Xpar(u2,:)];Xoff(i:(i+1),:)=xpar;

    if (u3<pc) then

     xoff=blcross(xpar);   //crossover

     Xoff(i:(i+1),:)=xoff;

    end

    for j=1:2

       u4=rand();

        if (u4<pm) then

          xoffl=mutate(xoff(j,:),gen,Ngen,xmin,xmax);  // mutation

          Xoff(i+j-1,:)=xoffl;

        end

    end

  end

end

[Xpar,bestpar]=evalplus1elitism(Xoff,Xoldpar,Npop,gen);  fmin=[fmin,bestpar($)];

//

//----------- results displays --------------

xset('window',0);

xbasc();

plot2d(0:Npop:(Npop*Ngen),fmin);

xtitle('convergence history','Neval','fmin');

xset('window',1);

xbasc();

plot2d1('onl',0:Npop:(Npop*Ngen),fmin);

xtitle('convergence history (log scale)','Neval','fmin');

//

disp('minimum obtained:');disp(bestpar(1:n));

disp('corresponding value by f:');disp(bestpar($));

disp('evaluation number:');disp(Npop*Ngen);

ANNEX 2 : computation of the reflectitivity spectrum of a FBG and of the errror function
//

function rhodot=f(z,rho,delta,lamb,zf,dn)   // right hand side of the ODE

d=splin(zf,dn);         // spline interpolation of the apodisation function

dnz=interp(z,zf,dn,d);

rhodot(1)=-2*delta*rho(2)-%pi*dnz*2*rho(1)*rho(2)/(lamb);

rhodot(2)=2*delta*rho(1)+%pi*dnz*(rho(1)^2-rho(2)^2)/(lamb)+%pi/lamb*dnz;

endfunction

//

//

Nc=evstr(x_dialog(' interpolation points number for the apodization','3'));

Np=evstr(x_dialog('number of computed points of the spectrum','80'));

dnmax=evstr(x_dialog('maximal value of the apodisation function','2E-4'));

//

n0=1.45;

L=0.01;

lambda0=1.55E-6;

lambdamin=1.5497E-6;

lambdamax=1.5503E-6;

//

z0=L;rho0=[0;0];

z=L:(-L/(Nc-1)):0;

zf=0:(L/(Nc-1)):L;

// random choice of the apodisation function

dn=dnmax*(ones(1,Nc)-2*rand(1,Nc));  

lambda=lambdamin:(lambdamax-lambdamin)/(Np-1):lambdamax;

rtarget=zeros(1,Np);

ref=[];

timer();

for i=1:Np

  delta0=2*n0*%pi*(1/(lambda(i))-1/lambda0);

  if abs(lambda(i)-lambda0)<0.1E-9 then rtarget(i)=1;end

  rho=ode(rho0,z0,z,list(f,delta0,lambda0,zf,dn));

  R=rho(1,$)^2+rho(2,$)^2;

  ref=[ref,R];

end

disp('computational time:');disp(timer());

xset('window',0);

xbasc();

d=splin(zf,dn);

dnz=interp(0:L/100:L,zf,dn,d);

plot2d(0:L/100:L,dnz)

plot2d(zf,dn,-1);

xtitle('apodisation function','z','dn(z)')

//

xset('window',1);

xbasc();

plot2d(lambda,ref)

plot2d(lambda,rtarget,2)

error=(sum((abs(ref-rtarget)).^0.75))^(1/0.75);

disp('error:');disp(error)

